

EUROPEAN ORGANIZATION FOR NUCLEAR RESEARCH

CERN-LEP/TH/83-30

THE MAD PROGRAM

by F C. Iselin

Presented at the 12th International Conference
on High-Energy Accelerators

August 11th - 16th 1983,
at Fermi National Laboratory, Batavia, Illinois.

Geneva, Switzerland
August 1983

CERN LIBRARIES, GENEVA



CM-P00047633

THE MAD PROGRAM

F Christoph Iselin
CERN, Geneva, Switzerland

Summary

The program "MAD" (Methodical Accelerator Design) has been written at CERN to provide a standard user interface for solving problems arising in accelerator design. Its aim is to solve these problems starting from a common input data structure, instead of having many programs each of which uses a completely different input format. To achieve this goal the program performs basic computations like computing Twiss parameters, chromaticities, or machine geometry; and it is able to feed other programs with the basic results. The program also provides flexible ways of matching various parameters of the machine.

MAD uses a format-free input language which is easy to learn and yet flexible enough to describe even very large accelerators in a few lines. When switching to a new problem on the same machine, the same input data is used. The new problem may just require a few lines of additional information.

Basic Computations Available in MAD

Twiss Parameters

MAD tracks the Twiss parameters for any arbitrary beam line. The initial conditions are normally given by the periodic solution. They may also be specified by the user. If a periodic beam line is symmetric, only half of it need to be entered, and the computing time is halved. The user is free to select any position(s) in the beam line where the Twiss parameters will be printed. In addition to the Twiss parameters, MAD also computes chromaticities, transition energy, as well as the largest values of β and dispersion for each plane. Optionally MAD also computes the chromatic functions¹².

The transfer matrices are computed by the same methods as in the program TRANSPORT³. Momentum-dependent effects are treated by using the second-order terms delivered by the TRANSPORT formalism. Should this prove inadequate, it is easy to plug in a different method.

Particle Tracking

At the time of writing, investigations are being made on the most appropriate methods for particle tracking to be added in MAD. It is planned to include tracking in the presence of synchrotron oscillations, and of field and positioning errors. Simulation of synchrotron radiation may also be provided.

Survey (Machine Geometry)

For large accelerators it is a tedious task to compute the position in cartesian space for all magnets. MAD computes these positions as well as the direction angles.

Definition Commands

When running the MAD program, the user enters a set of commands which are obeyed in the order read. Some commands serve to define beam elements:

```
DRIFT,D1,L=5
QUAD,QF,L=2,K1=-0 01345
SBEND,B,L=4,ANGLE=0 0113
```

The above commands define a drift space D1, a quadrupole QF, and a sector bending magnet B respectively. Blanks may be used freely in commands, and commands can continue on as many input lines as required. An ampersand (&) denotes continuation on the next line. Having defined a set of beam elements, the user may combine them with the LINE command to form sequences:

```
LINE,PER=(D1,B,D2,QF,D3,SF,D1,B,D2,QD,D3,SD)
```

This command defines a simple FODO cell. For more complex sequences, beam lines can be nested to any level and/or repeated:

```
LINE,INS1=( )
LINE,INS2=( )
LINE,SUP=(INS1,8*PER,INS2)
```

This defines a superperiod SUP consisting of 8 cells with the name PER, sandwiched between two insertions called INS1 and INS2. The definitions of INS1 and INS2 must of course be filled in.

Frequently the same beam line is used several times with small variations. Assume that in the above example the odd-numbered cells contain the sextupoles SF1 and SD1, while the even-numbered cells contain the sextupoles SF2 and SD2. To avoid defining two different cells, the sextupoles in the line PER are made formal arguments:

```
LINE,PER(SF,SD)=&
(D1,B,D2,QF,D3,SF,D1,B,D2,QD,D3,SD)
```

When this beam line is used, the arguments are replaced by the corresponding actual arguments:

```
LINE,SUP=&
(INS1,4*(PER(SF1,SD1),PER(SF2,SD2)),INS2)
```

This will be expanded to eight cells, as defined above.

Using the above facilities and suitable element definitions, the complete element sequence for the LEP machine can be written in 14 beam line definitions:

```

LINE,LEP=(4*SUP)
LINE,SUP=(OCT,-OCT)
LINE,OCT=(LOBS,RFS,DISS,ARC,DISL,RFL,LOBL)
LINE,LOBS=(L1,QS1,L2,QS2,L3,QS3,L4,QS4)
LINE,RFS=(L5,QS5,L5,QS6,L5,&
2*(QS7,L5,QS8,L5))
LINE,DISS=(QS11,L25,BW,L22,QS12,L25,B4,&
L22,QS13,L25,B4,L22,QS14,L25,B4,&
L31,QS15,L25,B4,L32,SF,L23,QS16)
LINE,ARC=(L21,B6,L22,SD,L23,QD,&
7*(CELL(SF1,SD1),CELL(SF,SD)),&
CELL(SF1,SD1),&
L24,B6,L41,QF,L21,B6,L22,SD4,L23,QD,&
7*(CELL(SF4,SD3),CELL(SF3,SD4)),&
CELL(SF4,SD3),&
L24,B6,L22,SF3,L23)
LINE,DISL=(QL16,L34,B4,L22,QL15,L33,B4,&
L22,QL14,L25,B4,L22,QL13,L25,B4,&
L22,QL12,L25,BW,L22,QL11)
LINE,RFL=(2*(L5,QL8,L5,QL7),&
L5,QL6,L5,QL5,L5)
LINE,LOBL=(QL4,L14,QL3,L13,QL2,L12,QL1,L11)
LINE,BW=(2*W,L26,2*W)
LINE,B4=(2*B,L26,2*B)
LINE,B6=(2*B,L26,2*B,L26,2*B)
LINE,CELL(SF,SD)=&
(L24,B6,L22,SF,L23,QF,&
L21,B6,L22,SD,L23,QD)

```

The beam lines BW, B4 and B6 represent strings of bending magnets. For preliminary calculations one will rather define them as single magnets. Having found a suitable design it is easy to replace them by the actual sequences.

Action Commands

Another set of commands requests specific actions on the machine structure. In order to do any computations, the user first specifies which beam line is to be studied. This is done by a command like

```
USE,OCT,SUPER=4,SYMM
```

All subsequent computations refer to the octant OCT, made symmetric by appending its mirror image, and repeated 4 times.

Twiss Parameters and Chromatic Functions

The actual computations are requested by a TWISS command. This command computes the Twiss parameters for several user-specified values of $\delta p/p$. By default the program prints results at both ends of the beam line. The user may however request additional positions for printing like

```
PRINT,QD[1/15]
PRINT,#27/30
```

These are requests for print-out after the 1st to 15th QD, and after the 27th to 30th physical element. As an option, MAD also prints the chromatic functions for the same beam line and the same values of $\delta p/p$. The print positions are selected by the same mechanism as for the Twiss parameters.

Survey

After a USE command the user may also enter a SURVEY command. This command causes MAD to print a table of positions and angles in cartesian space for all magnets in the accelerator. The print positions are selected by the same mechanism as for the TWISS command. The initial position and direction can be chosen freely.

Matching a Cell

Having defined a machine period PERIOD, one may match this beam line as a period:

```

CELL,PERIOD
VARY,QD[K1]
VARY,QF[K1]
CONSTR,#E,MUX=0.25,MUZ=0.16666667
MIGRAD,TOLER=1.0E-10
ENDMATCH

```

The CELL command initiates matching of a periodic structure. The strengths of the quadrupoles QF and QD are varied such as to obtain phase advances of one quarter and one sixth of 2π in the horizontal and vertical plane respectively at the end (position #E) of the period. The matching methods available are in essence the same as the ones used in the MINUIT program. The user makes his choice by entering e.g. MIGRAD or SIMPLEX. Matching mode is left by entering the ENDMATCH command.

Matching an Insertion

A simple insertion INSERT may be matched as follows:

```

MATCH,INSERT,LINE=CELL
CONSTR,#E,LINE=CELL1
VARY,Q1[K1]
! OTHER VARY COMMANDS REQUIRED
SIMPLEX
MIGRAD
ENDMATCH

```

The MATCH command initiates insertion matching mode and defines the beam line and initial conditions to be used. The above CONSTR command imposes conditions at the end (position #E). Thus the line INSERT is adjusted such that it fits between the two periodic lines CELL and CELL1. There is no need to enter numeric values for Twiss parameters at either end. The program computes these values and fills in the constraints. It is however permissible to enter numeric values instead of phrases like "LINE=CELL". Here the user wants to run first the SIMPLEX method, followed by the MIGRAD method. The ENDMATCH command terminates matching.

Sometimes it is desired to make part of an insertion look like a period, i.e. the Twiss parameters should be equal in two positions. This is requested by the command

```
COUPLE,M1,M2
```

where M1 and M2 identify the two positions. They may both be any names occurring in the beam line. The phase advances from M1 to M2 can also be fixed with this command.

Additional Facilities

Auxiliary Input

MAD has a mechanism to imbed predefined input files in its input stream. This allows to keep the accelerator structure in a data base, and to enter only the actions to be performed.

Ancillary Programs

A link is available to run the HARMON program^{5,6} as a subroutine. Any modified sextupole strengths will be replaced in the MAD data tables. All tables generated by MAD may be written out on coded disk files. A set of other programs is available to read these files and to perform additional calculations on the lattice. This idea had already been realized with the AGS program⁷. This program however writes the Twiss parameters in binary format. To allow easier transfer of results between different computers, MAD uses coded files.

An Example for Matching

A complete example for a matching run is shown below. Element definitions have been left out in order not to overload the example.

```
! DEFINE BEAM LINES
LINE, CELL=&
  (QDH, L24, B6, L22, SF, L23, QF, &
   L21, B6, L22, SD, L23, QDH)
MARKER, MARK
LINE, DISL=(QL16, L34, B4, L24, QL15, L33, B4, &
  L24, QL14, L25, B4, L24, QL13, L25, B4, &
  L24, QL12, L25, BW, L24, QL11)
LINE, RFL=(2*(L5, QL8, L5, QL7H, MARK, QL7H), &
  L5, QL6, L5, QL5, L5)
LINE, LOBL=&
  (QL4, L14, QL3, L13, QL2, L12, QL1, L11)
LINE, INSL=&
  (QDH, L24, B6, L22, SF3, L23, DISL, RFL, LOBL)
! ENTER MATCHING MODE
! DEFINE INITIAL VALUES
MATCH, INSL, LINE=CELL
! VARIABLE PARAMETERS
VARY, QL1[K1], STEP=0 001
VARY, QL2[K1], STEP=0 001
VARY, QL3[K1], STEP=0 001
VARY, QL4[K1], STEP=0 001
VARY, QL5[K1], STEP=0 001
VARY, QL6[K1], STEP=0 001
VARY, QL7[K1], STEP=0 001
VARY, QL8[K1], STEP=0 001
VARY, QL11[K1], STEP=0 001
VARY, QL12[K1], STEP=0 001
VARY, QL13[K1], STEP=0 001
VARY, QL14[K1], STEP=0 001
VARY, QL15[K1], STEP=0 001
VARY, QL16[K1], STEP=0 001
! REPLACE DEFAULT MATCHING WEIGHTS
WEIGHT, BETX=10, BETZ=50, ALFX=10, ALFZ=10, &
  MUX=10, MUZ=10, DX=10, DX'=100
! CONSTRAINTS FOR INTERACTION POINT
CONSTR, #E, BETX=3 2, BETZ=0 2, ALFX=0, ALFZ=0, &
  MUX=1 771875, MUZ=2 0125, DX=0, DX'=0
```

```
! CONSTRAINTS AT TWO INTERMEDIATE POINTS
! (MARK OCCURS TWICE IN RFL)
CONSTR, MARK, ALFX=0, ALFZ=0, DX=0, DX'=0
! LIMITS FOR DISPERSION IN BEAM LINE DISL
CONSTR, DISL, DX>0, DX<1 25
! PERFORM MATCH
MIGRAD, TOLER=1E-9
! LEAVE MATCHING MODE
ENDMATCH
```

How to get Access to MAD

The MAD program is written in standard FORTRAN 77. A copy of the source code is available from the author. The documentation consists of a reference manual⁸ and a primer⁹, both of which are available in print or in machine-readable form from the author.

MAD has been tested successfully on the CERN CDC 7600 computer as well as on the CERN IBM 3081 / Siemens 7880 system. The only restriction is that HARMON cannot be run on the CDC machine due to space limitations. It is hoped that this restriction can be overcome with future systems.

References

- 1 B Autin and A Verdier, Focusing Perturbations in Alternating Gradient Structures, CERN ISR-LTD/76-14
- 2 B Montague, Linear Optics for Improved Chromaticity Correction, CERN, LEP Note 165, 30 7 1979
- 3 K Brown, D C Carey, Ch Iselin and F Rothacker, TRANSPORT, a Program to Compute Charged Particle Beam Transport Systems, published simultaneously as CERN 73-16, FNAL 91, SLAC 91, revised as CERN 80-4
- 4 F James and M Roos, MINUIT, A Package of Programs to Minimize a Function of n Variables, Compute the Covariance Matrix, and Find the True Errors, CERN Program Library code D 506
- 5 M H R Donald, P L Morton and H Wiedemann, Proc Particle Accelerator Conference, Chicago, 1977, (IEEE Trans Nucl Sci NS-24, No 3 (1977) 1200)
- 6 M Donald and D Schofield, A User's Guide to the HARMON Program, CERN, LEP Note 420, 9 12 1982
- 7 E Keil, Y Marti, B W Montague and A Sudboe, AGS, The ISR Computer Program for Synchrotron Design, Orbit Analysis, and Insertion Matching, CERN 75-13, 1975
- 8 Ch Iselin, MAD, Methodical Accelerator Design, Reference Manual Available from the author
- 9 Ch Iselin, MAD, Methodical Accelerator Design, A Primer To be published